

# XLang

Max Heidinger, Pascal Riesinger

April 6, 2019

## 1 Einleitung

XLang ist eine Programmiersprache entwickelt im Rahmen der Vorlesung *Compilerbau* an der Dualen Hochschule Baden-Württemberg Karlsruhe.

## 2 Syntax

### 2.1 Grammatik

Die folgende Grammatik in der Extended-Backus-Naur-Form beschreibt die Syntax von XLang.

Startsymbol: Programm

```
Programm ::= NamenTeil '\n'
           TeilSeparator '\n'
           VariablenTeil '\n'
           TeilSeparator '\n'
           CodeTeil.
TeilSeparator ::= '===' { '=' }.
NamenTeil ::= { Zeichen }.
VariablenTeil ::= { Variable '\n' }.
Variable ::= [ImportExportFlag] VariablenName VariablenTyp.
ImportExportFlag ::= ImportFlag | ExportFlag.
ImportFlag ::= '->'.
ExportFlag ::= '<-'.
VariablenName ::= Buchstabe { Zeichen }.
VariablenTyp ::= 'int' | 'float' | 'string'.
CodeTeil ::= { Instruktion }.
Instruktion ::= (Zuweisung '\n') | Verzweigung | (Kommentar '\n').
Kommentar ::= '// ' { Zeichen }.
Zuweisung ::= VariablenName '=' EvaluationsWert.
EvaluationsWert ::= Wert
                  | Addition
                  | Subtraktion
                  | Multiplikation
                  | Division
                  | RestDivision.
Wert ::= VariablenName | Literal.
Literal ::= String | Int | Float.
Addition ::= EvaluationsWert '+' EvaluationsWert.
```

```

Subtraktion ::= EvaluationsWert '-' EvaluationsWert.
Multiplikation ::= EvaluationsWert '*' EvaluationsWert.
Division ::= EvaluationsWert '/' EvaluationsWert.
RestDivision ::= EvaluationsWert '%' EvaluationsWert.

Verzweigung ::= IfVerzweigung | WhileVerzweigung | ForVerzweigung.
InstruktionsBlock ::= '{' {Instruktion} '}'.

IfVerzweigung ::= 'if' Vergleich InstruktionsBlock
                {ElseIfVerzweigung} [ElseVerzweigung].
Vergleich ::= GleichheitsVergleich
            | UngleichheitsVergleich
            | KleinerVergleich
            | GroesserVergleich.
GleichheitsVergleich ::= EvaluationsWert '==' EvaluationsWert.
UngleichheitsVergleich ::= EvaluationsWert '!=' EvaluationsWert.
KleinerVergleich ::= EvaluationsWert '<' EvaluationsWert.
GroesserVergleich ::= EvaluationsWert '>' EvaluationsWert.
ElseIfVerzweigung ::= 'else if' Vergleich InstruktionsBlock.
ElseVerzweigung ::= 'else' InstruktionsBlock.

WhileVerzweigung ::= 'while' Vergleich InstruktionsBlock.
ForVerzweigung ::= 'for' Zuweisung ';' Vergleich ';' Zuweisung InstruktionsBlock.

Zeichen ::= Buchstabe | Ziffer
Buchstabe ::= 'A'|...| 'Z'| 'a'|...| 'z'.
AlleZeichen ::= Zeichen | Sonderzeichen.
Sonderzeichen ::= ' ' | '!' | ... | '@'.
Ziffer ::= '0'|...| '9'.
String ::= '''
        | ''' {AlleZeichen} '''
        | ''' {AlleZeichen} '''.
Int ::= Ziffer {Ziffer}.
Float ::= Int '.' Int.

```

## 2.2 Beschreibung

### 2.2.1 Struktur

Ein XLang Programm ist in 3 Teile aufgeteilt. Diese werden durch eine Folge von Gleichheitszeichen (=) voneinander abgetrennt, welche mindestens 3 Zeichen lang sein muss. Zunächst wird der Name des Programmes vermerkt. Anschließend folgt der Variablendeklarationsteil, in welchem alle im Programmteil verwendeten Variablen deklariert werden müssen. Der dritte Teil des Programmes ist der sogenannte Programmteil, welcher alle Instruktionen beinhaltet.

Leerzeichen können überall im Programm eingefügt werden. Zeilenenden werden durch ein einzelnes Linefeed-Zeichen (ASCII 0x0A) markiert. Diese Zeilenenden werden auch unter UNIX-ähnlichen Systemen verwendet.

Kommentare können im Codeteil nur in eigenen Zeilen mit // markiert verwendet werden.

### 2.2.2 Variablen

Es gibt in XLang drei Datentypen:

- Ganzzahlen, deklariert durch den Typ `int`.

- Gleitkommazahlen, deklariert durch den Typ `float`.
- Zeichenfolgen, deklariert durch den Typ `string`.

Wie bereits angemerkt müssen alle im Programm verwendeten Variablen im Variablenteil deklariert werden. Da XLang ohne Funktionen und Unterprogramme auskommt, müssen Variablen, welche als Eingabeparameter fungieren sollen, mit einer sogenannten `ImportFlag` gekennzeichnet werden. Eine Deklaration einer Eingabevariable sieht dann beispielsweise wie folgt aus: `-> name string`.

Die Eingabevariablen werden beim Programmstart aus der Kommandozeile ausgelesen. Dabei ist die Reihenfolge der Deklaration gleich der Reihenfolge der Übergabe.

Variablen, deren Wert am Ende des Programmes ausgegeben werden soll, müssen mit einer `ExportFlag` gekennzeichnet werden. Eine Deklaration einer Ausgabevariable sieht dann beispielsweise wie folgt aus: `<- ergebnis float`. Die Ausgabevariablen werden in der Reihenfolge ausgegeben, in welcher sie deklariert wurden.

Es gibt keine Möglichkeit, Variablen während der Ausführung auszugeben oder einzulesen.

Variablen können Literale, also konstante Werte oder Ausdrücke zugewiesen werden. Ein Ausdruck ist bei numerischen Werten entweder eine andere Variable oder eine mathematische Formel. Bei Rechnungen können die Grundrechenarten (Addition, Subtraktion, Multiplikation und Division) verwendet werden. Hierbei gilt die gleiche Priorisierung, wie im C-Standard.

Bei Zeichenketten ist ein Ausdruck entweder eine andere Variable oder eine Verkettung von mehreren Literalen und Variablen über den Additionsoperator.

### 2.2.3 Konditionen

Die Konditionalausdrücke in XLang folgen dem Beispiel anderer Programmiersprachen. Wenn der Vergleich nach dem `if` zu einem logisch "wahren" Wert evaluiert, werden alle Anweisungen im folgenden Anweisungsblock ausgeführt. Der `if`-Anweisung können optional beliebig viele `else if`-Anweisungen folgen. Diese werden nur ausgewertet, wenn der Vergleich der voranstehenden Anweisung zu einem unwahren Wert evaluiert wurde. Abgeschlossen werden kann ein Konditionalausdruck durch ein `else`. Die Anweisungen im `else`-Block werden ausgeführt, wenn kein vorheriger Vergleich im Konditionalausdruck zu einem wahren Wert evaluiert werden konnte.

### 2.2.4 Kommentare

Es gibt in XLang nur Zeilenkommentare. Diese werden mit doppelten Schrägstrichen eingeleitet und können jegliche Zeichen enthalten. Kommentare können im Codeteil vorkommen und enden mit dem Ende der Zeile. Kommentare werden vom Compiler ignoriert und haben keine Bedeutung für das Programm.

### 2.2.5 Schleifen

Es gibt in XLang zwei Arten von Schleifen. Eine `while`-Schleife führt den Codeblock so lange aus, bis die Bedingung im Kopf der Schleife zu einem unwahren Wert evaluiert wird. While-Schleifen sind kopfgesteuert.

Die zweite Art der Schleifen in XLang sind `for`-Schleifen. Diese erlauben es, eine Zuweisung im Fuß der Schleife durchzuführen. Zunächst wird eine Variable zugewiesen, welche dann in der Kondition in der Schleife verwendet wird. Der dritte Teil der Schleife weist der Variable einen neuen Wert zu. Vor jedem Durchlauf wird der Vergleich durchgeführt, dann der Programmblock, falls der Vergleich zu einem wahren Wert evaluiert, nach dem Ausführen des Programmblockes wird dann die Zuweisung ausgeführt.

## 3 Beispielprogramme

### 3.1 Berechnung der Fakultät

## Factorial Calculation

---

```
-> input_num int
<- result int
iterator int
```

---

```
result = 1
if input_num != 0 {
  for iterator = 2; iterator < input_num + 1; iterator = iterator + 1 {
    result = result * iterator
  }
}
```

## 3.2 Berechnung des größten gemeinsamen Teilers

Groesster Gemeinsamer Teiler

---

```
-> a int
-> b int
c int
<- ergebnis int
```

---

```
// Euklidischer Algorithmus
if a < b {
  c = b
  b = a
  a = c
}
```

```
c = 1
while c > 0 {
  c = a % b
  b = a / b
  if c > 0 {
    a = b
    b = c
  }
}
```

```
ergebnis = c
```